// File: contracts/interfaces/IERC20.sol



pragma solidity >=0.5.0;

interface IERC20 {
    event Approval(address indexed owner, address indexed spender, uint value);
    event Transfer(address indexed from, address indexed to, uint value);

    function name() external view returns (string memory);
    function symbol() external view returns (string memory);
    function decimals() external view returns (uint8);
    function totalSupply() external view returns (uint);
    function balanceOf(address owner) external view returns (uint);
    function allowance(address owner, address spender) external view returns (uint);

    function approve(address spender, uint value) external returns (bool);
    function transfer(address to, uint value) external returns (bool);
    function transferFrom(address from, address to, uint value) external returns (bool);
}

// File: contracts/interfaces/IKyotoxCallee.sol



pragma solidity >=0.5.0;

interface IKyotoxCallee {
    function KyotoxCall(address sender, uint amount0, uint amount1, bytes calldata data) external;
}

// File: contracts/interfaces/IKyotoxFactory.sol



pragma solidity >=0.5.0;

interface IKyotoxFactory {
    event PairCreated(address indexed token0, address indexed token1, address pair, uint);

    function getPair(address tokenA, address tokenB) external view returns (address pair);
    function allPairs(uint) external view returns (address pair);
    function allPairsLength() external view returns (uint);

    function createPair(address tokenA, address tokenB) external returns (address pair);

    function getOwner() external view returns(address);
    function getInitHash() external pure returns(bytes32);
    }

// File: contracts/interfaces/IKyotoxPair.sol



pragma solidity >=0.5.0;

interface IKyotoxPair {
    event Approval(address indexed owner, address indexed spender, uint value);
    event Transfer(address indexed from, address indexed to, uint value);

    function name() external pure returns (string memory);
    function symbol() external pure returns (string memory);
    function decimals() external pure returns (uint8);
    function totalSupply() external view returns (uint);
    function fee() external view returns (uint32);
    function balanceOf(address owner) external view returns (uint);
    function allowance(address owner, address spender) external view returns (uint);

    function approve(address spender, uint value) external returns (bool);
    function transfer(address to, uint value) external returns (bool);
    function transferFrom(address from, address to, uint value) external returns (bool);

    function DOMAIN_SEPARATOR() external view returns (bytes32);
    function PERMIT_TYPEHASH() external pure returns (bytes32);
    function nonces(address owner) external view returns (uint);

    function permit(address owner, address spender, uint value, uint deadline, uint8 v, bytes32 r, bytes32 s) external;

    event Mint(address indexed sender, uint amount0, uint amount1);
    event Burn(address indexed sender, uint amount0, uint amount1, address indexed to);
    event Swap(
        address indexed sender,
        uint amount0In,
        uint amount1In,
        uint amount0Out,
        uint amount1Out,
        address indexed to
    );
    event Sync(uint112 reserve0, uint112 reserve1);

    function MINIMUM_LIQUIDITY() external pure returns (uint);
    function factory() external view returns (address);
    function token0() external view returns (address);
    function token1() external view returns (address);
    function getReserves() external view returns (uint112 reserve0, uint112 reserve1, uint32 blockTimestampLast);
    function price0CumulativeLast() external view returns (uint);
    function price1CumulativeLast() external view returns (uint);
    function kLast() external view returns (uint);

    function mint(address to) external returns (uint liquidity);
    function burn(address to) external returns (uint amount0, uint amount1);
    function swap(uint amount0Out, uint amount1Out, address to, bytes calldata data) external;
    function skim(address to) external;
    function sync() external;

    function initialize(address, address, uint32, address) external;
}

// File: contracts/interfaces/KyotoxERC20.sol



pragma solidity >=0.5.0;

interface IKyotoxERC20 {
    event Approval(address indexed owner, address indexed spender, uint value);
    event Transfer(address indexed from, address indexed to, uint value);

    function name() external pure returns (string memory);
    function symbol() external pure returns (string memory);
    function decimals() external pure returns (uint8);
    function totalSupply() external view returns (uint);
    function balanceOf(address owner) external view returns (uint);
    function allowance(address owner, address spender) external view returns (uint);

    function approve(address spender, uint value) external returns (bool);
    function transfer(address to, uint value) external returns (bool);
    function transferFrom(address from, address to, uint value) external returns (bool);

    function DOMAIN_SEPARATOR() external view returns (bytes32);
    function PERMIT_TYPEHASH() external pure returns (bytes32);
    function nonces(address owner) external view returns (uint);

    function permit(address owner, address spender, uint value, uint deadline, uint8 v, bytes32 r, bytes32 s) external;
}

// File: contracts/KyotoxERC20.sol



pragma solidity >=0.5.16;

import './interfaces/KyotoxERC20.sol';
import './libraries/SafeMath.sol';

contract KyotoxERC20  {
    using SafeMath for uint;

    string public constant name = 'Kyotox - LP';
    string public constant symbol = 'KYO-LP';
    uint8 public constant decimals = 18;
    uint  public totalSupply;
    mapping(address => uint) public balanceOf;
    mapping(address => mapping(address => uint)) public allowance;

    bytes32 public DOMAIN_SEPARATOR;
    // keccak256("Permit(address owner,address spender,uint256 value,uint256 nonce,uint256 deadline)");
    bytes32 public constant PERMIT_TYPEHASH = 0x6e71edae12b1b97f4d1f60370fef10105fa2faae0126114a169c64845d6126c9;
    mapping(address => uint) public nonces;

    event Approval(address indexed owner, address indexed spender, uint value);
    event Transfer(address indexed from, address indexed to, uint value);

    constructor() public {
        uint chainId;
        assembly {
            chainId := chainid()
        }
        DOMAIN_SEPARATOR = keccak256(
            abi.encode(
                keccak256('EIP712Domain(string name,string version,uint256 chainId,address verifyingContract)'),
                keccak256(bytes(name)),
                keccak256(bytes('1')),
                chainId,
                address(this)
            )
        );
    }

    function _mint(address to, uint value) internal {
        totalSupply = totalSupply.add(value);
        balanceOf[to] = balanceOf[to].add(value);
        emit Transfer(address(0), to, value);
    }

    function _burn(address from, uint value) internal {
        balanceOf[from] = balanceOf[from].sub(value);
        totalSupply = totalSupply.sub(value);
        emit Transfer(from, address(0), value);
    }

    function _approve(address owner, address spender, uint value) private {
        allowance[owner][spender] = value;
        emit Approval(owner, spender, value);
    }

    function _transfer(address from, address to, uint value) private {
        balanceOf[from] = balanceOf[from].sub(value);
        balanceOf[to] = balanceOf[to].add(value);
        emit Transfer(from, to, value);
    }

    function approve(address spender, uint value) external returns (bool) {
        _approve(msg.sender, spender, value);
        return true;
    }

    function transfer(address to, uint value) external returns (bool) {
        _transfer(msg.sender, to, value);
        return true;
    }

    function transferFrom(address from, address to, uint value) external returns (bool) {
        if (allowance[from][msg.sender] != uint(-1)) {
            allowance[from][msg.sender] = allowance[from][msg.sender].sub(value);
        }
        _transfer(from, to, value);
        return true;
    }

    function permit(address owner, address spender, uint value, uint deadline, uint8 v, bytes32 r, bytes32 s) external {
        require(deadline >= block.timestamp, 'Kyotox: EXPIRED');
        bytes32 digest = keccak256(
            abi.encodePacked(
                '\x19\x01',
                DOMAIN_SEPARATOR,
                keccak256(abi.encode(PERMIT_TYPEHASH, owner, spender, value, nonces[owner]++, deadline))
            )
        );
        address recoveredAddress = ecrecover(digest, v, r, s);
        require(recoveredAddress != address(0) && recoveredAddress == owner, 'Kyotox: INVALID_SIGNATURE');
        _approve(owner, spender, value);
    }
}

// File: contracts/KyotoxFactory.sol



pragma solidity >=0.5.16;

import './interfaces/IKyotoxFactory.sol';
import './KyotoxPair.sol';

contract KyotoxFactory {
    address private owner;

    mapping(address => mapping(address => address)) public getPair;
    address[] public allPairs;

    uint32 public defaultFee = 50; // 0.5%

    mapping(address => bool) public canCreatePair;
    bool public freePairCreation = false;

    event PairCreated(address indexed token0, address indexed token1, address pair, uint);
    event OwnerChanged(address indexed oldowner, address indexed newOwner);

    event FreePairCreationUpdated(bool value);
    event CanCreatePairUpdated(address indexed account, bool value);

    event DefaultFeeUpdated(uint32 value);

    constructor(address _owner) public {
        require(_owner != address(0), 'Owner should be non-zero');
        owner = _owner;
    }

    modifier onlyOwner() {
        require(tx.origin == owner, 'Kyotox: NOT_ALLOWED');
        _;
    }

    modifier canCreate() {
        require(freePairCreation || canCreatePair[tx.origin], 'Kyotox: CREATE_NOT_ALLOWED');
        _;
    }

    function allPairsLength() external view returns (uint) {
        return allPairs.length;
    }

    function createPair(address tokenA, address tokenB) external canCreate returns (address pair) {
        require(tokenA != tokenB, 'Kyotox: IDENTICAL_ADDRESSES');
        (address token0, address token1) = tokenA < tokenB ? (tokenA, tokenB) : (tokenB, tokenA);
        require(token0 != address(0), 'Kyotox: ZERO_ADDRESS');
        require(getPair[token0][token1] == address(0), 'Kyotox: PAIR_EXISTS'); // single check is sufficient
        bytes memory bytecode = type(KyotoxPair).creationCode;
        bytes32 salt = keccak256(abi.encodePacked(token0, token1));
        assembly {
            pair := create2(0, add(bytecode, 32), mload(bytecode), salt)
        }
        IKyotoxPair(pair).initialize(token0, token1, defaultFee, owner); // fee sent to owner
        getPair[token0][token1] = pair;
        getPair[token1][token0] = pair; // populate mapping in the reverse direction
        allPairs.push(pair);
        emit PairCreated(token0, token1, pair, allPairs.length);
    }

    function getInitHash() public pure returns (bytes32) {
        bytes memory bytecode = type(KyotoxPair).creationCode;
        return keccak256(abi.encodePacked(bytecode));
    }

    function transferOwnership(address newOwner) external onlyOwner {
        require(newOwner != address(0), 'Owner should be non-zero');
        owner = newOwner;
        emit OwnerChanged(tx.origin, newOwner);
    }

    function getOwner() public view returns (address) {
        return owner;
    }

    /**
     * @notice Allows the contract owner to enable or disable free pair creation
     * @dev Only callable by the contract owner
     * @param value The new state of the free pair creation
     */
    function setFreePairCreation(bool value) external onlyOwner {
        require(freePairCreation != value, 'Kyotox: FREE_CREATION_ALREADY_SET');
        freePairCreation = value;
        emit FreePairCreationUpdated(value);
    }

    /**
     * @notice Allows the contract owner to enable or disable pair creation for an account
     * @dev Only callable by the contract owner
     * @param account The account for which to enable or disable pair creation
     * @param value The new state of the pair creation
     */
    function setCanCreatePair(address account, bool value) external onlyOwner {
        require(canCreatePair[account] != value, 'Kyotox: CREATE_PAIR_ALREADY_SET');
        canCreatePair[account] = value;
        emit CanCreatePairUpdated(account, value);
    }

    /**
     * @notice Allows the contract owner to set the default fee for new pairs
     * @dev Only callable by the contract owner
     * @param value The new default fee
     */
    function setDefaultFee(uint32 value) external onlyOwner {
        require(value <= 5_00, 'Kyotox: FEE_TOO_HIGH');
        defaultFee = value;
        emit DefaultFeeUpdated(value);
    }
}

// File: contracts/KyotoxPair.sol



pragma solidity >=0.5.16;

import './interfaces/IKyotoxPair.sol';
import './KyotoxERC20.sol';
import './libraries/Math.sol';
import './libraries/UQ112x112.sol';
import './interfaces/IERC20.sol';
import './interfaces/IKyotoxFactory.sol';
import './interfaces/IKyotoxCallee.sol';

contract KyotoxPair is KyotoxERC20 {
    using SafeMath for uint;
    using UQ112x112 for uint224;

    uint public constant MINIMUM_LIQUIDITY = 10 ** 3;
    uint public constant MAX_FEE = 5_00;
    uint public constant FEE_BASE = 100_00;
    bytes4 private constant SELECTOR = bytes4(keccak256(bytes('transfer(address,uint256)')));

    address public factory;
    address public token0;
    address public token1;

    uint112 private reserve0; // uses single storage slot, accessible via getReserves
    uint112 private reserve1; // uses single storage slot, accessible via getReserves
    uint32 private blockTimestampLast; // uses single storage slot, accessible via getReserves

    uint public price0CumulativeLast;
    uint public price1CumulativeLast;
    uint public kLast; // reserve0 * reserve1, as of immediately after the most recent liquidity event

    uint32 public fee;
    address public feeReceiver;
    uint public scheduledChange;

    uint private unlocked = 1;
    modifier lock() {
        require(unlocked == 1, 'Kyotox: LOCKED');
        unlocked = 0;
        _;
        unlocked = 1;
    }

    function getReserves() public view returns (uint112 _reserve0, uint112 _reserve1, uint32 _blockTimestampLast) {
        _reserve0 = reserve0;
        _reserve1 = reserve1;
        _blockTimestampLast = blockTimestampLast;
    }

    function _safeTransfer(address token, address to, uint value) private {
        (bool success, bytes memory data) = token.call(abi.encodeWithSelector(SELECTOR, to, value));
        require(success && (data.length == 0 || abi.decode(data, (bool))), 'Kyotox: TRANSFER_FAILED');
    }

    event Mint(address indexed sender, uint amount0, uint amount1);
    event Burn(address indexed sender, uint amount0, uint amount1, address indexed to);
    event Swap(
        address indexed sender,
        uint amount0In,
        uint amount1In,
        uint amount0Out,
        uint amount1Out,
        address indexed to
    );
    event Sync(uint112 reserve0, uint112 reserve1);
    event FeeParamsChanged(uint32 fee, address feeReceiver);

    constructor() public {
        factory = msg.sender;
    }

    modifier onlyOwner() {
        require(IKyotoxFactory(factory).getOwner() == msg.sender, 'Kyotox: FORBIDDEN');
        _;
    }

    // called once by the factory at time of deployment
    function initialize(address _token0, address _token1, uint32 _fee, address _feeReceiver) external {
        require(msg.sender == factory, 'Kyotox: FORBIDDEN'); // sufficient check
        token0 = _token0;
        token1 = _token1;
        fee = _fee;
        feeReceiver = _feeReceiver;
    }

    // update reserves and, on the first call per block, price accumulators
    function _update(uint balance0, uint balance1, uint112 _reserve0, uint112 _reserve1) private {
        require(balance0 <= uint112(-1) && balance1 <= uint112(-1), 'Kyotox: OVERFLOW');
        uint32 blockTimestamp = uint32(block.timestamp % 2 ** 32);
        uint32 timeElapsed = blockTimestamp - blockTimestampLast; // overflow is desired
        if (timeElapsed > 0 && _reserve0 != 0 && _reserve1 != 0) {
            // * never overflows, and + overflow is desired
            price0CumulativeLast += uint(UQ112x112.encode(_reserve1).uqdiv(_reserve0)) * timeElapsed;
            price1CumulativeLast += uint(UQ112x112.encode(_reserve0).uqdiv(_reserve1)) * timeElapsed;
        }
        reserve0 = uint112(balance0);
        reserve1 = uint112(balance1);
        blockTimestampLast = blockTimestamp;
        emit Sync(reserve0, reserve1);
    }

    // this low-level function should be called from a contract which performs important safety checks
    function mint(address to) external lock returns (uint liquidity) {
        (uint112 _reserve0, uint112 _reserve1, ) = getReserves(); // gas savings
        uint balance0 = IERC20(token0).balanceOf(address(this));
        uint balance1 = IERC20(token1).balanceOf(address(this));
        uint amount0 = balance0.sub(_reserve0);
        uint amount1 = balance1.sub(_reserve1);

        uint _totalSupply = totalSupply;
        if (_totalSupply == 0) {
            liquidity = Math.sqrt(amount0.mul(amount1)).sub(MINIMUM_LIQUIDITY);
            _mint(address(0), MINIMUM_LIQUIDITY); // permanently lock the first MINIMUM_LIQUIDITY tokens
        } else {
            liquidity = Math.min(amount0.mul(_totalSupply) / _reserve0, amount1.mul(_totalSupply) / _reserve1);
        }
        require(liquidity > 0, 'Kyotox: INSUFFICIENT_LIQUIDITY_MINTED');
        _mint(to, liquidity);

        _update(balance0, balance1, _reserve0, _reserve1);
        emit Mint(msg.sender, amount0, amount1);
    }

    // this low-level function should be called from a contract which performs important safety checks
    function burn(address to) external lock returns (uint amount0, uint amount1) {
        (uint112 _reserve0, uint112 _reserve1, ) = getReserves(); // gas savings
        address _token0 = token0; // gas savings
        address _token1 = token1; // gas savings
        uint balance0 = IERC20(_token0).balanceOf(address(this));
        uint balance1 = IERC20(_token1).balanceOf(address(this));
        uint liquidity = balanceOf[address(this)];

        uint _totalSupply = totalSupply;
        amount0 = liquidity.mul(balance0) / _totalSupply; // using balances ensures pro-rata distribution
        amount1 = liquidity.mul(balance1) / _totalSupply; // using balances ensures pro-rata distribution
        require(amount0 > 0 && amount1 > 0, 'Kyotox: INSUFFICIENT_LIQUIDITY_BURNED');
        _burn(address(this), liquidity);
        _safeTransfer(_token0, to, amount0);
        _safeTransfer(_token1, to, amount1);
        balance0 = IERC20(_token0).balanceOf(address(this));
        balance1 = IERC20(_token1).balanceOf(address(this));

        _update(balance0, balance1, _reserve0, _reserve1);
        emit Burn(msg.sender, amount0, amount1, to);
    }

    function _feeAmount(uint amount) internal view returns (uint feeAmount) {
        feeAmount = (amount * fee) / (FEE_BASE - fee);
    }

    function _totalAmount(uint amount) internal view returns (uint totalAmount) {
        totalAmount = amount + _feeAmount(amount);
    }

    // this low-level function should be called from a contract which performs important safety checks
    function swap(uint amount0Out, uint amount1Out, address to, bytes calldata data) external lock {
        require(amount0Out > 0 || amount1Out > 0, 'Kyotox: INSUFFICIENT_OUTPUT_AMOUNT');
        (uint112 _reserve0, uint112 _reserve1, ) = getReserves(); // gas savings

        require(
            _totalAmount(amount0Out) < _reserve0 && _totalAmount(amount1Out) < _reserve1,
            'Kyotox: INSUFFICIENT_LIQUIDITY'
        );

        uint balance0;
        uint balance1;

        {
            // scope for _token{0,1}, avoids stack too deep errors
            address _token0 = token0;
            address _token1 = token1;
            require(to != _token0 && to != _token1, 'Kyotox: INVALID_TO');

            if (amount0Out > 0) {
                // assume that enough amountIn was provided to cover both fee and amountOut
                _safeTransfer(_token0, feeReceiver, _feeAmount(amount0Out)); // optimistically transfer fee
                _safeTransfer(_token0, to, amount0Out); // optimistically transfer tokens
            }
            if (amount1Out > 0) {
                // assume that enough amountIn was provided to cover both fee and amountOut
                _safeTransfer(_token1, feeReceiver, _feeAmount(amount1Out)); // optimistically transfer fee
                _safeTransfer(_token1, to, amount1Out); // optimistically transfer tokens
            }
            if (data.length > 0) IKyotoxCallee(to).KyotoxCall(msg.sender, amount0Out, amount1Out, data);
            balance0 = IERC20(_token0).balanceOf(address(this));
            balance1 = IERC20(_token1).balanceOf(address(this));
        }

        uint amount0In = balance0 > _reserve0 - _totalAmount(amount0Out)
            ? balance0 - (_reserve0 - _totalAmount(amount0Out))
            : 0;
        uint amount1In = balance1 > _reserve1 - _totalAmount(amount1Out)
            ? balance1 - (_reserve1 - _totalAmount(amount1Out))
            : 0;

        require(amount0In > 0 || amount1In > 0, 'Kyotox: INSUFFICIENT_INPUT_AMOUNT');
        {
            // scope for reserve{0,1}Adjusted, avoids stack too deep errors
            uint balance0Adjusted = balance0.mul(1000).sub(amount0In.mul(3));
            uint balance1Adjusted = balance1.mul(1000).sub(amount1In.mul(3));
            require(
                balance0Adjusted.mul(balance1Adjusted) >= uint(_reserve0).mul(_reserve1).mul(1000 ** 2),
                'Kyotox: K'
            );
        }

        _update(balance0, balance1, _reserve0, _reserve1);
        emit Swap(msg.sender, amount0In, amount1In, amount0Out, amount1Out, to);
    }

    // force balances to match reserves
    function skim(address to) external lock {
        address _token0 = token0; // gas savings
        address _token1 = token1; // gas savings
        _safeTransfer(_token0, to, IERC20(_token0).balanceOf(address(this)).sub(reserve0));
        _safeTransfer(_token1, to, IERC20(_token1).balanceOf(address(this)).sub(reserve1));
    }

    // force reserves to match balances
    function sync() external lock {
        _update(IERC20(token0).balanceOf(address(this)), IERC20(token1).balanceOf(address(this)), reserve0, reserve1);
    }

    function scheduleFeeChange() external onlyOwner {
        scheduledChange = block.timestamp + 86_400;
    }

    function setFeeData(uint32 _fee, address _feeReceiver) external onlyOwner {
        require(_feeReceiver != address(0), 'Kyotox: ZERO_FEE_RECEIPIENT');
        require(_fee < MAX_FEE, 'Kyotox: FEE_TOO_LARGE');
        require(scheduledChange < block.timestamp, 'Kyotox: TIMEOUT_NOT_REACHED');
        require(scheduledChange > 0, 'Kyotox: ACTION_NOT_SCHEDULED');
        scheduledChange = 0;
        fee = _fee;
        feeReceiver = _feeReceiver;
        emit FeeParamsChanged(fee, feeReceiver);
    }
}

// File: contracts/libraries/Math.sol



pragma solidity >=0.5.16;

// a library for performing various math operations

library Math {
    function min(uint x, uint y) internal pure returns (uint z) {
        z = x < y ? x : y;
    }

    // babylonian method (https://en.wikipedia.org/wiki/Methods_of_computing_square_roots#Babylonian_method)
    function sqrt(uint y) internal pure returns (uint z) {
        if (y > 3) {
            z = y;
            uint x = y / 2 + 1;
            while (x < z) {
                z = x;
                x = (y / x + x) / 2;
            }
        } else if (y != 0) {
            z = 1;
        }
    }
}

// File: contracts/libraries/SafeMath.sol



pragma solidity =0.6.6;

// a library for performing overflow-safe math, courtesy of DappHub (https://github.com/dapphub/ds-math)

library SafeMath {
    function add(uint x, uint y) internal pure returns (uint z) {
        require((z = x + y) >= x, 'ds-math-add-overflow');
    }

    function sub(uint x, uint y) internal pure returns (uint z) {
        require((z = x - y) <= x, 'ds-math-sub-underflow');
    }

    function mul(uint x, uint y) internal pure returns (uint z) {
        require(y == 0 || (z = x * y) / y == x, 'ds-math-mul-overflow');
    }
}

// File: contracts/libraries/UQ112x112.sol



pragma solidity >=0.5.16;

// a library for handling binary fixed point numbers (https://en.wikipedia.org/wiki/Q_(number_format))

// range: [0, 2**112 - 1]
// resolution: 1 / 2**112

library UQ112x112 {
    uint224 constant Q112 = 2**112;

    // encode a uint112 as a UQ112x112
    function encode(uint112 y) internal pure returns (uint224 z) {
        z = uint224(y) * Q112; // never overflows
    }

    // divide a UQ112x112 by a uint112, returning a UQ112x112
    function uqdiv(uint224 x, uint112 y) internal pure returns (uint224 z) {
        z = x / uint224(y);
    }
}